# Maintainer: Justin Davis <jrcd83@gmail.com>
# $Id: PKGBUILD 86612 2013-03-20 17:14:47Z eric $

pkgname=perl-dist-checkconflicts
pkgver=0.02
pkgrel=1
pkgdesc="declare version conflicts for your dist"
arch=(any)
license=(PerlArtistic GPL)
options=(!emptydirs)
depends=('perl-list-moreutils>=0.12' 'perl-sub-exporter')
makedepends=('perl-extutils-makemaker>=6.31')
checkdepends=('perl-test-fatal' 'perl-test-simple')
url=https://metacpan.org/release/Dist-CheckConflicts
source=("http://search.cpan.org/CPAN/authors/id/D/DO/DOY/Dist-CheckConflicts-${pkgver}.tar.gz")
md5sums=(64b8d342ba11336b2969c274a60bbc5f)
sha512sums=(e3a342141b13e3604f226d53706fa50b6dc9f96c39ad9c83b074289aff852cbbab97bdf7d389d426116f609589e3d0282b033e79230a5a4ed8782e2b0e58426d)
_distdir="${srcdir}/Dist-CheckConflicts-${pkgver}"

build() {
  ( export PERL_MM_USE_DEFAULT=1 PERL5LIB=""                 \
      PERL_AUTOINSTALL=--skipdeps                            \
      PERL_MM_OPT="INSTALLDIRS=vendor"     \
      PERL_MB_OPT="--installdirs vendor" \
      MODULEBUILDRC=/dev/null

    cd "$_distdir"
    /usr/bin/perl Makefile.PL
    make
  )
}

check() {
  ( export PERL_MM_USE_DEFAULT=1 PERL5LIB=""
    cd "$_distdir"
    make test
  )
}

package() {
    cd "$_distdir"
    make DESTDIR="$pkgdir" install
}

# Local Variables:
# mode: shell-script
# sh-basic-offset: 2
# End:
# vim:set ts=2 sw=2 et:
